# $Id: PKGBUILD 184379 2013-05-06 19:37:20Z foutrelis $
# Maintainer: Eric Belanger <eric@archlinux.org>

pkgname=gd
pkgver=2.0.36RC1
pkgrel=6
pkgdesc="Library for the dynamic creation of images by programmers"
arch=('i686' 'x86_64')
url="http://www.libgd.org/"
license=('custom')
depends=('libpng' 'fontconfig' 'libjpeg')
optdepends=('perl: bdftogd script')
options=('!libtool')
source=("ftp://ftp.archlinux.org/other/${pkgname}/${pkgname}-${pkgver}.tar.bz2"
        'libpng14.patch')
md5sums=('e876979ca3130623a4e7866d9579171a'
         '7921556f4960e8d66cf4fb67ea7101b5')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  patch -Np1 -i ${srcdir}/libpng14.patch
  ./configure --prefix=/usr --without-xpm
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -D -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
